
========BSIM_NMOS_Gate_Cap============

HOW TO DO CAPACITIVE MODEL VERIFICATION.  

BSIM_NMOS_Gate_Cap
*

*

.OPTIONS  GMIN=1e-15    METHOD=gear    ABSTOL=1e-15    temp=27
VIN       IN     0      DC     0V      AC    1mV
RG        IN     A      10k
VG        B      2      DC     5V 
VG2       2      0      DC     -2V      
MG        B      A      B      B       NMOSC   W=3u     L=1u AD=7p AS=7p PD=10u PS=10u

.control
setplot new
compose VbiasVals   values  0 .25 .5 .75 1 1.25 1.5 1.75 2 2.25 2.5 2.75 3 3.25 3.5 3.75 4
settype voltage     VbiasVals

let NoOfVbias =     length(VbiasVals)
* ==================Loop_Vbias==============
echo               "=>Vtest = Cgate="
set thisName =      CV_NMOS_F
let $thisName =     0 * vector(NoOfVbias)
let k =             1

repeat              17
*while              (k <= NoOfVbias )
let Vbias =         VbiasVals[k-1]



alter VG dc =       $&Vbias
ac                  dec 30 .0001GHz 100GHz

let g =             (db(a)+60)
let f =             mag(frequency[ sortorder( abs(g+3.0) )[0] ] )
let cg =            1/(6.28e4*f)
echo               "$&unknown.Vbias   $&cg"
let                 unknown.{$thisName}[unknown.k-1] = cg
destroy
let k =             k + 1
*endwhile
endrepeat

settype notype     unknown.{$thisName}
let Vgate =        VbiasVals -2
setscale           Vgate
plot               $thisName   title "CV_Curve_Nmos"
echo              "Done."
.endc

.model             NMOSC              NMOS
+ Level=  8        Tnom=27.0
*------------------Process----------------------------------------------- 
+ tox=160e-10      xj=0.25e-06        nch=0.5e+17                        
*------------------V_threshold------------------------------------------- 
+ vth0=0.72        nlx=0.12e-06                                          
*------------------Bulk-------------------------------------------------- 
+ k1=1.04          k2=-1.209E-01
*------------------mobility----------------------------------------------
+ u0=678           ua=8.964e-10                                          
+ ub=1.472e-18     uc=-4.441E-17      vsat=86000                         
*------------------Subthresshold-----------------------------------------
+ nfactor=1.8                                                               
+ cit=-5.0E-04     voff=-7.862E-02                                        
+ eta0=4.441e-16   etab=-2.E-01       dsub=0.7                           
*------------------Hot electrons----------------------------------------- 
+ alpha0=1.61e-05  beta0=36.68                                           
*------------------VAF---------------------------------------------------
+ lint=.12e-06     pclm=.19           pscbe1=3.79e+08   pscbe2=9.4e-05   
+ delta=0.01655    pvag=0.4484                                            
*------------------Bulk_diode-------------------------------------------- 
+ js=5.858e-08                                                              
*------------------Resistance-------------------------------------------- 
+ rsh=70           rdsw=375                                                
+ wr=0.7586        prwb=0             prwg=-4.441E-17                    
*------------------Capacitance-------------------------------------------
+ cj=0.0002424     cjsw=2.73e-10      mj=0.3551           mjsw=0.3873    
+ cgso=9e-13       cgdo=9e-13         cgbo=7e-10                         
+ cdsc=-2.4E-4     cdscd=-1.506E-01   cdscb=-2.219E-04                   
+ pb=0.5614        pbsw=0.8           xpart=0                             
+ dlc=5e-08        dwc=1.5e-07  
*------------------BulkChargeEffect--------------------------------------  
* a0=0.7           a1=0                a2=1                ags=0.05583    
* b0=6.305e-08     b1=6.579e-08        keta=-1.531E-02                    
*------------------ShortChannel------------------------------------------ 
+ dvt0=2.2         dvt1=0.53          dvt2=-1.521E-01      drout=0.76       
+ pdiblcb=.4       pdiblc1=0.00886    pdiblc2=0.00029                      
*------------------NarrowChannel-----------------------------------------
+ w0=2.6e-04       wint=0.16e-06                                          
+ ww=-9.525E-14    wwn=1.0                                               
+ dvt0w=0          dvt1w=5.3e6        dvt2w=-1.E-01                        
+ k3=2.53          k3b=-5             dwg=0                dwb=0  
*------------------Noise-------------------------------------------------
* af=1             kf=1e-28           ef=0.95                            
*------------------Temperature------------------------------------------- 
* pvsat=0          ute=-1.258E+00     kt1=-3.85E-01                      
* kt1l=0           kt2=-3.098E-02     ua1=5.705e-09                      
* ub1=-1.147E-17   uc1=-1.302E-01     at=20380                           
* prt=-3.287E+02   lk1=0              lk2=0                              
* lvsat=0          la0=0              lags=0             lute=0          



* luc=0                                                                                                   
.end

*=====================END_OF_SPICE=======================

